

# NAG Library Routine Document

## F07JDF (DPTTRF)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07JDF (DPTTRF) computes the modified Cholesky factorization of a real  $n$  by  $n$  symmetric positive definite tridiagonal matrix  $A$ .

### 2 Specification

```
SUBROUTINE F07JDF (N, D, E, INFO)
  INTEGER          N, INFO
  REAL (KIND=nag_wp) D(*), E(*)
```

The routine may be called by its LAPACK name *dpstrf*.

### 3 Description

F07JDF (DPTTRF) factorizes the matrix  $A$  as

$$A = LDL^T,$$

where  $L$  is a unit lower bidiagonal matrix and  $D$  is a diagonal matrix with positive diagonal elements. The factorization may also be regarded as having the form  $U^T D U$ , where  $U$  is a unit upper bidiagonal matrix.

### 4 References

None.

### 5 Parameters

- 1: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 2: D(\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the dimension of the array D must be at least  $\max(1, N)$ .  
*On entry:* must contain the  $n$  diagonal elements of the matrix  $A$ .  
*On exit:* is overwritten by the  $n$  diagonal elements of the diagonal matrix  $D$  from the  $LDL^T$  factorization of  $A$ .
- 3: E(\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the dimension of the array E must be at least  $\max(1, N - 1)$ .  
*On entry:* must contain the  $(n - 1)$  subdiagonal elements of the matrix  $A$ .  
*On exit:* is overwritten by the  $(n - 1)$  subdiagonal elements of the lower bidiagonal matrix  $L$ . (E can also be regarded as containing the  $(n - 1)$  superdiagonal elements of the upper bidiagonal matrix  $U$ .)

4: INFO – INTEGER

*Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0 and INFO < N

The leading minor of order  $\langle value \rangle$  is not positive definite, the factorization could not be completed.

INFO > 0 and INFO = N

The leading minor of order  $n$  is not positive definite, the factorization was completed, but  $D(N) \leq 0$ .

## 7 Accuracy

The computed factorization satisfies an equation of the form

$$A + E = LDL^T,$$

where

$$\|E\|_{\infty} = O(\epsilon)\|A\|_{\infty}$$

and  $\epsilon$  is the *machine precision*.

Following the use of this routine, F07JEF (DPTTRS) can be used to solve systems of equations  $AX = B$ , and F07JGF (DPTCON) can be used to estimate the condition number of  $A$ .

## 8 Parallelism and Performance

Not applicable.

## 9 Further Comments

The total number of floating-point operations required to factorize the matrix  $A$  is proportional to  $n$ .

The complex analogue of this routine is F07JRF (ZPTTRF).

## 10 Example

This example factorizes the symmetric positive definite tridiagonal matrix  $A$  given by

$$A = \begin{pmatrix} 4.0 & -2.0 & 0 & 0 & 0 \\ -2.0 & 10.0 & -6.0 & 0 & 0 \\ 0 & -6.0 & 29.0 & 15.0 & 0 \\ 0 & 0 & 15.0 & 25.0 & 8.0 \\ 0 & 0 & 0 & 8.0 & 5.0 \end{pmatrix}.$$

## 10.1 Program Text

```

Program f07jdfc

!      F07JDF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
      Use nag_library, Only: dpttrf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: info, n
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: d(:), e(:)
!      .. Executable Statements ..
      Write (nout,*) 'F07JDF Example Program Results'
      Write (nout,*)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n

      Allocate (d(n),e(n-1))

!      Read the lower bidiagonal part of the tridiagonal matrix A from
!      data file

      Read (nin,*) d(1:n)
      Read (nin,*) e(1:n-1)

!      Factorize the tridiagonal matrix A

!      The NAG name equivalent of dpttrf is f07jdf
      Call dpttrf(n,d,e,info)

      If (info>0) Then
         Write (nout,99999) 'The leading minor of order ', info, &
           ' is not positive definite'
      End If

!      Print details of the factorization

      Write (nout,*) 'Details of factorization'
      Write (nout,*)
      Write (nout,*) ' The diagonal elements of D'
      Write (nout,99998) d(1:n)
      Write (nout,*)
      Write (nout,*) ' Sub-diagonal elements of the Cholesky factor L'
      Write (nout,99998) e(1:n-1)

99999 Format (1X,A,I3,A)
99998 Format (1X,8F9.4)
      End Program f07jdfc

```

## 10.2 Program Data

```

F07JDF Example Program Data
  5                               :Value of N
  4.0  10.0  29.0  25.0   5.0 :End of diagonal D
 -2.0  -6.0  15.0   8.0       :End of sub-diagonal E

```

### 10.3 Program Results

F07JDF Example Program Results

Details of factorization

The diagonal elements of D

4.0000 9.0000 25.0000 16.0000 1.0000

Sub-diagonal elements of the Cholesky factor L

-0.5000 -0.6667 0.6000 0.5000

---